The cookingsymbols package*

Sven Harder
sven_one1@gmx.de

November 17, 2011

1 Introduction

This package includes 11 symbols. They are original created, when I was searching for symbols for typesetting recipes. However, I didn't find any symbols, so I decided to create my own ones. This is the result ;)

At the end of creating these symbols, I realized that METAFONT is out of date. But at this time, I had no time to create the same symbols as vector based fonts. This is still a task for the future.

2 Symbols

The following symbols are created by this package and are available by these macros. **Important:** The symbols are enhanced for a better view (by \Large).

Macro	Symbol
\Oven	
\Topbottomheat	
\Topheat	
\Bottomheat	
\Fanoven	
\Gasstove	
\Dish	
\Knife	
\Fork	
\Spoon	
\Gloves	

And now have fun with these new symbols!

*This document corresponds to cookingsymbols v1.0, dated 2011/11/02.

3 Implementation

3.1 cookingsymbols.sty

The following content can be found in the derived file cookingsymbols.sty. It includes the macros for an easy access and the definition of the (new) font.

```
⟨*package⟩

\DeclareFontFamily{U}{cookingsymbols}{}

\DeclareFontShape{U}{cookingsymbols}{m}{n}{<-> cookingsymbols}{}
\DeclareFontShape{U}{cookingsymbols}{bx}{n}{<-> sub cookingsymbols/m/n }{}

\newcommand{\Oven}{{\usefont{U}{cookingsymbols}{m}{n}\symbol{0}}}
\newcommand{\Topbottomheat}{{\usefont{U}{cookingsymbols}{m}{n}\symbol{1}}}
\newcommand{\Topheat}{{\usefont{U}{cookingsymbols}{m}{n}\symbol{2}}}
\newcommand{\Bottomheat}{{\usefont{U}{cookingsymbols}{m}{n}\symbol{3}}}
\newcommand{\Fanoven}{{\usefont{U}{cookingsymbols}{m}{n}\symbol{4}}}
\newcommand{\Gasstove}{{\usefont{U}{cookingsymbols}{m}{n}\symbol{5}}}
\newcommand{\Dish}{{\usefont{U}{cookingsymbols}{m}{n}\symbol{6}}}
\newcommand{\Knife}{{\usefont{U}{cookingsymbols}{m}{n}\symbol{7}}}
\newcommand{\Fork}{{\usefont{U}{cookingsymbols}{m}{n}\symbol{8}}}
\newcommand{\Spoon}{{\usefont{U}{cookingsymbols}{m}{n}\symbol{9}}}
\newcommand{\Gloves}{{\usefont{U}{cookingsymbols}{m}{n}\symbol{10}}}

⟨/package⟩
```

3.2 cookingsymbols.mf

The following content can be found in the derived file cookingsymbols.mf. In this file the symbols are defined by using METAFONT.

```
⟨*mf⟩

mode_setup;

font_coding_scheme = "cookingsymbols for recipes";
font_identifier = "cookingsymbols";

u#:=0.68pt#;
define_pixels(u);

font_size 10pt#;
%design_size=10pt#;
linewidth=0.4pt;

thinpen.w := 0.5linewidth;
pen normalpen, thinpen;
normalpen := pencircle scaled 1linewidth;
thinpen := pencircle scaled thinpen.w;
```

3.2.1 Oven

```
%% oven symbol
"Oven";
beginchar(0,10u#,10u#,0);
  pickup pencircle scaled 0.75 linewidth;

  % Umrandung
  y1=y2; y3=y4;
  x1=x4; x2=x3;
  x2-x1=w;
  y2-y3=h;
  z4=(0u, 0u);
  draw z1--z2--z3--z4--cycle;

  % Ofenklappe
  r:=1u; % Radius
  z5=z4+(r+0.7u, 0.7u);
  x5=x8; x6=x7;
  x9=x12; x10=x11;
  y5=y12; y6=y11;
  y7=y10; y8=y9;

  z6=z5+(-r, r);
  z8=z7+(r,r);
  z10=z9+(r,-r);
  x11-x6=w-1.4u; % Breite
  y8-y5=6.3u; % Höhe

  fill z5{left}..z6{up}..z7{up}..z8{right}..z9{right}..z10{down}..z11{down}..z12{left}..cycle;

  % Ofengriff
  b:=3.2u; % Breite vom Griff
  h1:=0.5u; % Höhe
  z14=z8+((x9-x8)/2-b/2, -0.8u);
  x14=x13; x15=x16;
  x16=x13+b;
  y14=y13+h1;
  y15=y14;
  y16=y13;

  unfill z13--z14--z15--z16--cycle;

  % Display
  b:=2.8u; % Breite vom Display
  h1:=1.05u;   % Höhe
  z17=z1+(0.8u, -1u);
  x18=x19; x20=x17;
```

```
  y18=y17; y19=y20;
  x18-x17=b;
  y18-y19=h1;

  fill z17--z18--z19--z20--cycle;

  % Knöpfe
  r:=0.48u; % Radius der Knöpfe
  b:=1.3u; % Abstand der Knöpfe untereinander
  h1:=y19+(y18-y19)/2;
  x:=x18+0.8u;
  x21=x23; y24=y22;
  z24=(x,h1);
  z23=(x+r,h1-r);
  x22=x24+2*r;
  y21=y23+2*r;
  %fill z21..z22..z23..z24..cycle;

  for i=0 upto 3:
    fill (z21+(i*1b,0))..(z22+(i*b,0))..(z23+(i*b,0))..(z24+(i*b,0))..cycle;
  endfor

  %labels(range 1 thru 24);
endchar;
```

3.2.2 Top and bottom heat

```
%% top and bottomheat symbol
"Topbottomheat";
beginchar(1,11u#,10u#,0);
  pickup normalpen;
  % Umrandung
  ra:=1.0u; % Radius der Umrandung
  y1=y4; y2=y3;
  y5=y8; y6=y7;
  x1=x8; x2=x7;
  x3=x6; x4=x5;

  z2=z1+(ra,ra);
  z4=z3+(ra,-ra);
  z8=z7+(-ra,ra);
  x4-x1=w;
  y2-y7=h;
  z8=(0,ra);

  draw z1{up}..z2{right}..z3{right}..z4{down}..z5{down}..z6{left}..z7{left}..z8{up}..cycle;

  % Stäbe
  pickup pensquare scaled 1.15linewidth;
  % Oberhitze
```

```
  xa:=1.4u; % Abstand zum Rand
  ya:=1.85u;
  z9=(xa, (y2-y7)-ya);
  x10=(x4-x1)-xa;
  y10=y9;
  draw z9..z10;

  % Unterhitze
  x11=x9;x12=x10;
  y11=ya;
  y12=y11;
  draw z11..z12;

  %labels(range 1 thru 12);

endchar;
```

3.2.3 Top heat

This is nearly the same definition as Topbottomheat, except there is only one rod.

```
%% top heat symbol
"Topheat";
beginchar(2,11u#,10u#,0);
  pickup normalpen;
  % Umrandung
  ra:=1.0u; % Radius der Umrandung
  y1=y4; y2=y3;
  y5=y8; y6=y7;
  x1=x8; x2=x7;
  x3=x6; x4=x5;

  z2=z1+(ra,ra);
  z4=z3+(ra,-ra);
  z8=z7+(-ra,ra);
  x4-x1=w;
  y2-y7=h;
  z8=(0,ra);

  draw z1{up}..z2{right}..z3{right}..z4{down}..z5{down}..z6{left}..z7{left}..z8{up}..cycle;

  % Stäbe
  pickup pensquare scaled 1.15linewidth;
  % Oberhitze
  xa:=1.4u; % Abstand zum Rand
  ya:=1.85u;
  z9=(xa, (y2-y7)-ya);
  x10=(x4-x1)-xa;
  y10=y9;
  draw z9..z10;
```

```
180
181 endchar;
182
```

3.2.4 Bottom heat

This is nearly the same definition as Topbottomheat, except there is only one rod.

```
183 %% bottom heat symbol
184 "Bottomheat";
185 beginchar(3,11u#,10u#,0);
186   pickup normalpen;
187   % Umrandung
188   ra:=1.0u; % Radius der Umrandung
189   y1=y4; y2=y3;
190   y5=y8; y6=y7;
191   x1=x8; x2=x7;
192   x3=x6; x4=x5;
193
194   z2=z1+(ra,ra);
195   z4=z3+(ra,-ra);
196   z8=z7+(-ra,ra);
197   x4-x1=w;
198   y2-y7=h;
199   z8=(0,ra);
200
201   draw z1{up}..z2{right}..z3{right}..z4{down}..z5{down}..z6{left}..z7{left}..z8{up}..cycle;
202
203   % Stäbe
204   pickup pensquare scaled 1.15linewidth;
205   % Oberhitze
206   xa:=1.4u; % Abstand zum Rand
207   ya:=1.85u;
208   z9=(xa, (y2-y7)-ya);
209   x10=(x4-x1)-xa;
210   y10=y9;
211
212   % Unterhitze
213   x11=x9;x12=x10;
214   y11=ya;
215   y12=y11;
216   draw z11..z12;
217
218 endchar;
219
```

3.2.5 Fanoven

```
220 %% fanoven symbol
221 "Fanoven";
```

```
beginchar(4,11u#,10u#,0);
  pickup normalpen;
  % Umrandung
  ra:=1.0u; % Radius der Umrandung
  y1=y4; y2=y3;
  y5=y8; y6=y7;
  x1=x8; x2=x7;
  x3=x6; x4=x5;

  z2=z1+(ra,ra);
  z4=z3+(ra,-ra);
  z8=z7+(-ra,ra);
  x4-x1=w;
  y2-y7=h;
  z8=(0,ra);

  draw z1{up}..z2{right}..z3{right}..z4{down}..z5{down}..z6{left}..z7{left}..z8{up}..cycle;

  % Propeller
  z15=(w/2,(h/2)-1.05u); % Rotationspunkt
  a:=1.3; % Skalierungsfaktor
  z9=z15+a*(0.95u,2.66u);
  %z9=z15+(0.8u,3u);
  y10=y9;
  x10=w-x9;
  z11=z9 rotatedaround(z15,120);
  z12=z10 rotatedaround(z15,120);
  z13=z9 rotatedaround(z15,-120);
  z14=z10 rotatedaround(z15,-120);

  z16=z15 + a*(0,4u);
  z17=z16 rotatedaround(z15,120);
  z18=z16 rotatedaround(z15,-120);

  fill z9{dir 90}..z16{dir 180}..z10{dir 270}..z13{dir -30}..z18{dir 60}..z14{dir 150}..z11{dir 21

  %labels(range 1 thru 18);
endchar;
```

3.2.6 Gasstove

```
%% gasstove symbol
"Gasstove";
beginchar(5,11u#,10u#,0);
  pickup normalpen;
  % Umrandung
  ra:=1.0u; % Radius der Umrandung
  y1=y4; y2=y3;
  y5=y8; y6=y7;
  x1=x8; x2=x7;
```

```
  x3=x6; x4=x5;

  z2=z1+(ra,ra);
  z4=z3+(ra,-ra);
  z8=z7+(-ra,ra);
  x4-x1=w;
  y2-y7=h;
  z8=(0,ra);

  draw z1{up}..z2{right}..z3{right}..z4{down}..z5{down}..z6{left}..z7{left}..z8{up}..cycle;

  % Flamme
  z9=(w/2,1.0u); % Ursprung
  z10=(w/2,9u);  % Endpunkt
  z11=z9+(-1.4u, 3.1u);
  x12=w-x11;y12=y11;
  %z12=z9+(1.5u, 3.5u);

  fill z9{dir 135}..z11{up}..z10{dir 70}--cycle;
  %fill z9{dir 145}..z10{dir 65}--cycle;

  fill z9{dir 45}..z12{up}..z10{dir 110}--cycle;

  % Innere Flamme
  z13=z9+(0,0.7u); % Urspung
  z14=z13+(0, 3.7u); % Endpunkt
  z15=z9+(-0.5u,2.1u);
  z16=z9+(0.5u, 2.1u);

  unfill z13{dir 130}..z15{up}..z14{dir 65}--cycle;

  unfill z13{dir 60}..z16{up}..z14{dir 115}--cycle;

  %labels(range 1 thru 16);

endchar;
```

3.2.7 Dish

This is the definition of two circles (with different radii).

```
%% dish symbol
"Dish";
beginchar(6,10u#,10u#,0);
  pickup normalpen;
  ra:=0.5h;
  x1=x3;y4=y2;
  z4=(0,ra);
  z3=(ra,0);
  x2=x4+2*ra;
```

```
316   y1=y3+2*ra;
317   draw z1..z2..z3..z4..cycle;
318
319   ri:=3.7u;
320   x5=x7;y8=y6;
321   z8=(ra-ri,ra);
322   z7=(ra,ra-ri);
323   y6=0.5*(y5-y7) + (ra-ri);
324   x5=0.5*(x6-x8) + (ra-ri);
325   draw z5..z6..z7..z8..cycle;
326
327 %  labels(range 1 thru 8);
328 endchar;
329
```

3.2.8 Knife

```
330 %% knife symbol
331 "Knife";
332 beginchar(7,1.7u#,10u#,0);
333   pickup normalpen;
334   b:=0.8u; % Breite des Stiels
335   z1=(w, 0u);
336   x2=x1-b; y2=y1;
337
338   z5=(x2,4.8u);
339   z6=z5+(-0.85u, 2u);
340   x7=x1; y7=y1+10u;
341
342
343   fill z1--z2--z5{dir 150}..z6{up}..z7--cycle;
344
345   %labels(range 1 thru 11);
346
347 endchar;
348
```

3.2.9 Fork

```
349 %% fork symbol
350 "Fork";
351 beginchar(8,2u#,10u#,0);
352   pickup normalpen;
353
354   b:=0.8u;  % Breite des Stiels
355   zb:=0.2u; % Zackenbreite
356   za=0.4u;  % Zackenabstand
357   z1=(w/2+b/2, 0u);
358   x2=x1-b; y2=y1;
359   x3=x2; y3=y2+6u;
360   x4=x1; y4=y3;
```

```
361
362   fill z1--z2--z3--z4--cycle;
363
364   x5=x3-(4*zb+3*za)/2 +b/2;
365   y5=y3+1.9u;
366   x6=x5; y6=y5+2.1u;
367   x7=x6+zb; y7=y6;
368   x8=x7; y8=y5;
369
370   x9=x8+za; y9=y8;
371   x10=x9; y10=y6;
372   x11=x10+zb; y11=y10;
373   x12=x11; y12=y9;
374
375   x13=x12+za; y13=y12;
376   x14=x13; y14=y6;
377   x15=x14+zb; y15=y14;
378   x16=x15; y16=y5;
379
380   x17=x16+za; y17=y16;
381   x18=x17; y18=y6;
382   x19=x18+zb; y19=y18;
383   x20=x19; y20=y5;
384
385   fill z3{dir 130}..z5{up}--z6--z7--z8--z9--z10--z11--z12--z13--z14--z15--z16--z17--z18--z19--z20{
386
387   %labels(range 1 thru 20);
388
389 endchar;
390
```

3.2.10 Spoon

```
391 %% spoon symbol
392 "Spoon";
393 beginchar(9, 3.4u#, 10u#,0);
394   pickup normalpen;
395
396   b:=0.8u;  % Breite des Stiels
397   % Stiel
398   z1=(w/2+b/2, 0u);
399   x2=x1-b; y2=y1;
400   x3=x2; y3=y2+5.74u;
401   x4=x1; y4=y3;
402
403   fill z1--z2--z3--z4--cycle;
404
405   % Oberteil
406   x5=x3-1.0u;
407   x6=x4+(x3-x5);
408   y5=y6=y3+(y7-y3)/2 - 0.2u; % halbe Breite
```

```
409   z7=(x3+(x4-x3)/2, y3+4u); % y: maximale Ausdehnung
410
411   draw z7{dir -170}..z5{down}..(x3+(x4-x3)/2, y3){dir -10};
412   draw (x3+(x4-x3)/2, y3){dir 10}..z6{up}..z7{dir 170};%..(z7-(b/2, 0));
413
414   %labels(range 1 thru 12);
415 endchar;
416
```

3.2.11 Gloves

```
417 %% gloves symbol
418 "Gloves";
419 beginchar(10,10.4u#,10u#,0);
420   pickup thinpen;
421
422   a:=1.05; % Skalierungsfaktor
423   breite:=a*4.3u; % Breite unten am Handschuh
424   hoehe:=a*1.8u;
425
426   z5=a*(3.3u-0.4u, 3u); % Rotationspunkt 1. Handschuh
427   z9=z5 + a*(-0.55u, 0.1u); % Rotationspunkt 2. Handschuh
428   alpha:=-37; % Rotationswinkel -37
429   beta:=-26;  % Verschiebung des zweiten Handschuhs bzgl. des oberen
430
431   % Oberer Handschuh
432   z4=(0.8u, 1u);
433   x1=x4; x2=x3;
434   y1=y2; y3=y4;
435   x2-x1=breite;
436   y1-y4=hoehe;
437
438   fill z1--z2--z3--z4--cycle rotatedaround(z5, alpha);
439
440   % Daumen
441   z7=z1 + a*(0.5thinpen.w+0.2u, 3.5u);
442   z6=z7 + a*(-1.2u, 1.3u);
443
444   draw ((z1+(0.5thinpen.w,0)){up}..z6{dir 75}..z7{dir -80}) rotatedaround(z5, alpha);
445
446   % Oberer Rand/Begrenzung
447   z8=z1 + ((x2-x1)/2 + a*0.2u, a*8.2u);
448
449   draw (z7{dir 90}..z8{right}..(z2+(-0.5thinpen.w,0)){dir -90}) rotatedaround(z5, alpha);
450
451
452
453   % Zweiter Handschuh
454   fill z1--(z1-(-a*0.8u, (y1-y4)/2+a*0.6u))--z4--cycle rotatedaround(z9, alpha+beta);
455
456   fill z4--z2--z3--cycle rotatedaround(z9, alpha+beta);
```

```
457
458   %z10=z1 + a*(1u, 7.55u); % Schnittpunkt der beiden Handschuhe
459   z10=z1 + a*(1u, 7.75u);
460
461   draw (z10..z8{right}..z2{dir -90}) rotatedaround(z9, alpha+beta);
462
463
464
465   % 'Weier' Trennstrich
466   z11=z3 - (0, 0.5thinpen.w);
467   x12=x1; y12=y11;
468   unfill (z4--z3--z11--z12--cycle) rotatedaround(z5, alpha);
469
470
471   %labels(range 1 thru 20);
472 endchar;
473
474 end
475 ⟨/mf⟩
```

Index

Numbers written in italic refer to the page where the corresponding entry is described; numbers underlined refer to the code line of the definition; numbers in roman refer to the code lines where the entry is used.